

Portable Program Review

Vol. 1, No. 4

October, 1985

THE HAUNTED RAM

Do you get the feeling that your laptop is watching your every move? Does each beep sound ominous? Does the low battery light blink some secretive Morse Code? Even now, is the cassette recorder preparing to strike? If so -- don't play Haunted House. See Page.....

2

COME BACK IN SIX MONTHS

Certificates of deposit, or CDs, are great investments. Basically, you give your bank so much money, wait out the term of the CD, and retrieve the principal with interest. Usually, the term is a multiple of six months. Here's a program to make calculating the exact due date easier. See Page.....

5

TIME FOR WORK

The Tandy 200 is well-suited to life in an office or classroom -- places where meetings and schedules are commonplace. So, when the laptop's not busy taking notes or telecommunications, why not use it as an appointment secretary? See Page.....

6

HAUNTED ROM

Haunted House is a menu-driven adventure game for the Model 100 and Tandy 200. The objectives are clear: Explore every room of the old mansion and collect all treasure in sight. Easy? But watch out for traps!

Our thanks to SSG David W. Stewart for this excellent program.

Listing of HAUNT.BA, the Haunted House.

```
1000 CLS
:PRINT "*****"
1010 PRINT "*** HAUNT, PRODUCED BY ***"
:PRINT "*** David W. Stewart ***"
1020 PRINT "*****"
1030 FOR Z=1 TO 1000
:NEXT Z
:BEEP
1040 CLS
:GOTO 2960
1050 CLEAR 1000
:CLS
1060 FOR Z=1 TO VAL(RIGHT$(TIME$,2))
:I=RND(1)
:NEXT Z
1070 CLS
:CALL 17001
1080 PRINT @ 125,"THINKING, PLEASE
WAIT"
:CALL 17006
1090 DIM A(15),B(15),C(15),D(15),E(15),
Z(15)
1100 FOR Z=1 TO 15
:Z(Z)=0
:NEXT Z
1110 FOR Z=1 TO 15
1120 I=INT(RND(1)*15+1)
1130 IF Z(I)=1 THEN 1120
1140 A(Z)=I
:Z(I)=1
1150 NEXT Z
1160 FOR Z=1 TO 15
:Z(Z)=0
:NEXT Z
1170 FOR Z=1 TO 15
1180 I=INT(RND(1)*15+1)
1190 IF Z(I)=1 THEN 1180
1200 B(Z)=I
:Z(I)=1
```

```
1210 IF B(Z)=A(Z) THEN Z=100
:GOTO 1160
1220 NEXT Z
1230 FOR Z=1 TO 15
:Z(Z)=0
:NEXT Z
1240 FOR Z=1 TO 15
1250 I=INT(RND(1)*15+1)
1260 IF Z(I)=1 THEN 1250
1270 C(Z)=I
:Z(I)=1
1280 IF C(Z)=A(Z)ORC(Z)=B(Z) THEN Z=100
:GOTO 1230
1290 NEXT Z
1300 FOR Z=1 TO 15
:Z(Z)=0
:NEXT Z
1310 FOR Z=1 TO 15
1320 I=INT(RND(1)*15+1)
1330 IF Z(I)=1 THEN 1320
1340 D(Z)=I
:Z(I)=1
1350 NEXT Z
1360 FOR Z=1 TO 15
:Z(Z)=0
:NEXT Z
1370 FOR Z=1 TO 15
1380 I=INT(RND(1)*15+1)
1390 IF Z(I)=1 THEN 1380
1400 E(Z)=I
:Z(I)=1
1410 NEXT Z
1420 CLS
1430 LINE(10,11)-(10,41)
1440 LINE(10,11)-(14,15)
1450 LINE(10,41)-(14,37)
1460 LINE(14,15)-(14,37)
1470 LINE(14,20)-(19,20)
1480 LINE(19,20)-(19,32)
1490 LINE(19,32)-(14,32)
1500 LINE(40,11)-(40,41)
1510 LINE(40,11)-(36,15)
1520 LINE(36,15)-(36,37)
1530 LINE(36,37)-(40,41)
1540 LINE(31,20)-(36,20)
1550 LINE(31,20)-(31,32)
1560 LINE(31,32)-(36,32)
1570 LINE(19,20)-(31,32)
1580 LINE(19,32)-(31,20)
1590 LINE(10,11)-(40,11)
1600 LINE(10,41)-(40,41)
1610 FOR Z=1 TO 15
:IF D(Z)=1 THEN A=Z
```



```

1620 NEXT Z
1630 GOSUB 1760
1640 GOSUB 2240
1650 IF L=2 THEN GOSUB 2420
1660 PRINT @ 88,"1. MOVE    4. DROP
1670 PRINT @128,"2. LOOK    5. INVENTORY
1680 PRINT @168,"3. TAKE    6. SUICIDE
1690 PRINT @ 48,"TREASURES =";T;
1700 A$=INKEY$
      :IF A$="" THEN 1700
1710 IF VAL(A$)=0 THEN 1700
1720 PRINT @ 0,"";
1730 PRINT @ 164," ";
1740 ON VAL(A$) GOSUB 2460,2420,2530,
      2710,2720,2770
1750 GOTO 1630
1760 PRINT @ 241,"";
1770 PRINT @ 281,"";
1780 ON E(A) GOTO 1790,1820,1850,1880,
      1910,1940,1970,2000,2030,2060,
      2090,2120,2150,2180,2210
1790 PRINT @ 242,"DUSTY";
1800 PRINT @ 282,"ATTIC";
1810 RETURN
1820 PRINT @ 242,"MOLDY";
1830 PRINT @ 281,"KITCHEN";
1840 RETURN
1850 PRINT @ 241,"BLOODY";
1860 PRINT @ 281,"BEDROOM";
1870 RETURN
1880 PRINT @ 242,"DIRTY";
1890 PRINT @ 281,"CLOSET";
1900 RETURN
1910 PRINT @ 242,"DAMP";
1920 PRINT @ 281,"BASEMENT";
1930 RETURN
1940 PRINT @ 241,"SMELLY";
1950 PRINT @ 281,"HALLWAY";
1960 RETURN
1970 PRINT @ 241,"CREEPY";
1980 PRINT @ 281,"LIVINGROOM";
1990 RETURN
2000 PRINT @ 241,"SECRET";
2010 PRINT @ 281,"PASSAGE";
2020 RETURN
2030 PRINT @ 242,"OLD";
2040 PRINT @ 281,"LIBRARY";
2050 RETURN
2060 PRINT @ 241,"BUTLER'S";
2070 PRINT @ 282,"ROOM";
2080 RETURN
2090 PRINT @ 241,"MAID'S";
2100 PRINT @ 282,"ROOM";

```

```

2110 RETURN
2120 PRINT @ 241,"MYSTERY";
2130 PRINT @ 282,"ROOM";
2140 RETURN
2150 PRINT @ 241,"CLUTTERED";
2160 PRINT @ 281,"GARAGE";
2170 RETURN
2180 PRINT @ 242,"DARK";
2190 PRINT @ 281,"DUNGEON";
2200 RETURN
2210 PRINT @ 242,"MOLDY";
2220 PRINT @ 281,"BATHROOM";
2230 RETURN
2240 IF D(A)=1 THEN O=175
      :B$="KEY"
2250 IF D(A)=2 THEN O=131
      :B$="FLINT & STEEL"
2260 IF D(A)=3 THEN O=152
      :B$="TORCH"
2270 IF D(A)=4 THEN O=145
      :B$="BOTTLE OF LIQUID"
2280 IF D(A)=5 THEN O=128
      :B$="PHONE"
2290 IF D(A)=6 THEN O=169
      :B$="EGYPTIAN RING"
2300 IF D(A)=7 THEN O=144
      :B$="FERTILITY STATUE"
2310 IF D(A)=8 THEN O=132
      :B$=" TO Y CAR"
2320 IF D(A)=9 THEN O=171
      :B$="COLUMBUS'S DIARY"
2330 IF D(A)=10 THEN O=168
      :B$="GOLD CROSS"
2340 IF D(A)=11 THEN O=170
      :B$="STAMP ALBUM"
2350 IF D(A)=12 THEN O=157
      :B$="DIAMOND"
2360 IF D(A)=13 THEN O=134
      :B$="PANDORA'S BOX"
2370 IF D(A)=14 THEN O=167
      :B$="TRAP DOOR"
2380 IF D(A)=15 THEN O=143
      :B$="EXIT"
2390 IF D(A)=0 THEN O=32
      :B$=""
2400 PRINT @ 0,"";
2410 RETURN
2420 '
2430 PRINT @ 1,B$
2440 PRINT @ 164,CHR$(O);
2450 RETURN
2460 A$=INKEY$
      :SOUND1,1
      :FOR Z=1 TO 100

```



```

: NEXT Z
: IF A$="" THEN 2460
2470 FOR Z=1 TO 15
: IF D(Z)=1 THEN GOTO 1630
2480 NEXT Z
2490 IF A$="1" THEN A=A(A)
2500 IF A$="2" THEN A=B(A)
2510 IF A$="3" THEN A=C(A)
2520 RETURN
2530 PRINT @ 164," ";
2540 IF D(A)=1 THEN C$=C$+B$+" "
: D(A)=0
2550 IF D(A)=2 THEN L=L+1
: C$=C$+B$+" "
: D(A)=0
2560 IF D(A)=3 THEN L=L+1
: C$=C$+B$+" "
: D(A)=0
2570 IF D(A)=4 THEN GOTO 2780
2580 IF D(A)=5 THEN PRINT @ 1,
"THE PHONE IS OUT OF ORDER"
: FOR Z=1 TO 1000
: NEXT Z
2590 IF D(A)=6 THEN T=T+1
: D(A)=0
: C$=C$+B$+" "
2600 IF D(A)=7 THEN T=T+1
: D(A)=0
: C$=C$+B$+" "
2610 IF D(A)=8 THEN T=T+1
: D(A)=0
: C$=C$+B$+" "
2620 IF D(A)=9 THEN T=T+1
: D(A)=0
: C$=C$+B$+" "
2630 IF D(A)=10 THEN T=T+1
: D(A)=0
: C$=C$+B$+" "
2640 IF D(A)=11 THEN T=T+1
: D(A)=0
: C$=C$+B$+" "
2650 IF D(A)=12 THEN T=T+1
: D(A)=0
: C$=C$+B$+" "
2660 IF D(A)=13 THEN GOTO 2830
2670 IF D(A)=14 THEN GOTO 2860
2680 IF D(A)=15 AND T=7 THEN 2890
2690 IF D(A)=15 THEN D(A)=0
2700 RETURN
2710 T=0
: C$=""
: RETURN
2720 FOR Z=1 TO LEN(C$)
2730 PRINT @ 75,MID$(C$,Z,1);

```

```

2740 FOR I=1 TO 100
: NEXT I
2750 NEXT Z
2760 GOTO 1630
2770 CLS
: SOUND 15000,10
: FOR Z=1 TO 1000
: NEXT Z
: GOTO 1050
2780 CLS
2790 PRINT @ 125,"YOU DRANK THE LIQUID
& DIED!"
2800 FOR Z=1 TO 1000
: NEXT
2810 BEEP
2820 FOR Z=1 TO 2000
: NEXT
: GOTO 1050
2830 CLS
2840 PRINT @ 122,"YOU OPENED PANDORA'S
BOX AND DIED"
2850 FOR Z=1 TO 2000
: NEXT Z
: BEEP
: GOTO 1050
2860 CLS
2870 PRINT @ 120,"YOU FELL THROUGH THE
TRAP DOOR AND DIED"
2880 FOR Z=1 TO 2000
: NEXT Z
: BEEP
: GOTO 1050
2890 CLS
: PRINT @ 120,C$
2900 PRINT
2910 PRINT "YOU HAVE WON ALL OF THE
TREASURES AND
2920 PRINT "ESCAPED THE HAUNTED HOUSE.
2930 PRINT "CONGRATULATIONS!!!!
2940 FOR Z=16000 TO 500 STEP -500
: SOUNDZ,1
: NEXT Z
2950 FOR Z=1 TO 2000
: NEXT Z
: GOTO 1050
2960 PRINT
2970 PRINT "YOU'RE IN A HAUNTED HOUSE."
2980 PRINT "TRY TO GET OUT WITH ALL OF"
2990 PRINT "THE TREASURES. GOOD LUCK!"
3000 PRINT "WHEN YOU HEAR FOOTSTEPS,
(LISTEN NOW),"
3010 PRINT "PRESS 1, 2, OR 3 TO CHOOSE"
3020 PRINT "A DOOR TO ENTER."
3030 PRINT "* PRESS (hold down) ANY KEY

```



```

    TO BEGIN *";
3040 A$=INKEY$
    :I=INT(RND(1)*100)
3050 SOUND1,20
3060 IF A$="" THEN 3040 ELSE 1050 □

```

SIX MONTHS

Louis C. Graue of Bowling Green, Ohio, takes the credit for SIXMO.BA. Thanks to him, we're that much richer -- at least in terms of software.-Ed.

Two common financial problems that have been solved by computers: finding number of days between dates, and determining a future date that's so many days away.

Another problem: finding a future date so many months away. That, in part, is what SIXMO.BA does.

Why would this information be necessary? Many bonds and other financial papers are issued for a given number of months. It's important to know when these bonds mature, so that they can be re-invested as quickly as possible.

The SIXMO.BA program calculates bond coupon dates, as well as the accrued interest. All that's required to run the program are the bond's maturity date, face value, coupon rate and the settlement rate.

SIXMO.BA can be dissected to extract the financial and calendar routines. Using the pieces, specialized programs can be created.

Listing for SIXMO.BA by Louis C. Graue.

```

1000 DIM LD(12)
1010 FOR I=1 TO 12
    :READ LD(I)
    :NEXT I
1020 DATA 31, 28, 31, 30, 31, 30, 31,
    31, 30, 31, 30, 31
1030 CLS

```

```

1040 PRINT "                                BOND ACCRUAL"
1050 PRINT
1060 PRINT "PLEASE ENTER THE MATURITY
    DATE:"
1070 INPUT "MONTH (1 - 12)           ==>";MM
1080 INPUT "DAY (1 - 31)             ==>";DM
1090 INPUT "YEAR (ALL 4 DIGITS) ==>";YM
1100 PRINT
1110 PRINT "PLEASE ENTER THE SETTLEMENT
    DATE:"
1120 INPUT "MONTH (1 - 12)           ==>";MS
1130 INPUT "DAY (1 - 31)             ==>";DS
1140 INPUT "YEAR (ALL 4 DIGITS) ==>";YS
1150 PRINT
1160 INPUT "BOND FACE VALUE           ==>";FV
1170 INPUT "BOND COUPON RATE % ==>";CR
1180 GOSUB 1590
1190 '
1200 'GET DAYS BETWEEN LAST COUPON AND
    SETTLEMENT
1210 M1=ML
    :D1=DL
    :Y1=YL
1220 GOSUB 1480
1230 LC=DZ
1240 M1=MS
    :D1=DS
    :Y1=YS
1250 GOSUB 1480
1260 LS=DZ-LC
1270 '
1280 'DAYS BETWEEN LAST & NEXT COUPON
1290 M1=MN
    :D1=DN
    :Y1=YN
1300 GOSUB 1480
1310 LN=DZ-LC
1320 '
1330 'ACCRUED INTEREST
1340 AI=.005*CR*FV*LS/LN
1350 CLS
1360 PRINT "                                BOND ACCRUAL"
1370 YL$=RIGHT$(STR$(YL),2)
1380 YN$=RIGHT$(STR$(YN),2)
1390 PRINT
1400 PRINT "LAST COUPON DATE ==> ";
    ML;"/";DL;"/";YL$
1410 PRINT "NEXT COUPON DATE ==> ";
    MN;"/";DN;"/";YN$
1420 PRINT USING "ACCRUED INTEREST
    ==>$$$###.##";AI
1430 PRINT

```



```

1440 INPUT "Do you wish to enter
      another bond (Y/N)";A$
1450 IF A$="Y" OR A$="y" THEN 1030
      ELSE END
1460 '
1470 'DAYS TO ZERO
1480 LEAP=ABS((Y1 MOD 4 = 0) AND
      ((Y1 MOD 100 <> 0) OR (Y1 MOD 400
      = 0)))
1490 DZ=D1
1500 IF M1=1 THEN 1540
1510 FOR I=1 TO M1-1
1520 DZ=DZ+LD(I)
1530 NEXT I
1540 DZ=DZ+Y1*365+Y1\4+1-Y1\100+Y1\400
1550 IF LEAP=1 AND M1<3 THEN DZ=DZ-1
1560 RETURN
1570 '
1580 'LAST AND NEXT COUPON DATES
1590 MR=MM
1600 YR=YS
1610 DR=DM
1620 IF MM=2 AND DM=28 AND YM MOD 4 <>
      0 AND YR MOD 4 = 0 THEN DR=29
1630 IF MM=2 AND DM=29 AND YM MOD 4 = 0
      AND YR MOD 4 <> 0 THEN DR=28
1640 IF MS+6<MR OR (MS+6=MR AND DS<=DR)
      THEN GOSUB 1850
1650 IF MR+6<MS OR (MR+6=MS AND DR<=DS)
      OR (MR+6=MS AND DR=31 AND DS=30)
      THEN GOSUB 1950
1660 IF MS<MR OR (MS=MR AND DS<DR) THEN
      1750
1670 MN=MR
1680 DN=DR
1690 YN=YR
1700 GOSUB 1850
1710 ML=MR
1720 DL=DR
1730 YL=YR
1740 GOTO 1820
1750 MN=MR
1760 DN=DR
1770 YN=YR
1780 GOSUB 1850
1790 ML=MR
1800 DL=DR
1810 YL=YR
1820 RETURN
1830 '
1840 'SIX MONTHS EARLIER
1850 MX=MR
      :DX=DR
      :YX=YR

```

```

1860 IF MR>6 THEN MX=MR-6
1870 IF MR<=6 THEN MX=MR+6
      :YX=YR-1
1880 DX=LD(MX)
      :IF DX>DR THEN DX=DR
1890 IF DR=LD(MR)+ABS((MR=2)AND(YR MOD
      4 = 0)) THEN DX=LD(MX)
1900 IF YX MOD 4 = 0 AND MX=2 AND DR>28
      THEN DX=29
1910 DR=DX
      :MR=MX
      :YR=YX
1920 RETURN
1930 '
1940 'SIX MONTHS LATER
1950 MX=MR
      :DX=DR
      :YX=YR
1960 IF MR>6 THEN MX=MR-6
      :YX=YR+1
1970 IF MR<=6 THEN MX=MR+6
1980 DX=LD(MX)
      :IF DX>DR THEN DX=DR
1990 IF DR=LD(MR)+ABS((MR=2)AND(YR MOD
      4 = 0)) THEN DX=LD(MX)
2000 IF YX MOD 4 = 0 AND MX=2 AND DR>28
      THEN DX=29
2010 DR=DX
      :MR=MX
      :YR=YX
2020 RETURN □

```

WORK TIME

Our thanks to Dr. Leo
Finkelstein, Jr., for the
timely ACLOCK.BA program. -Ed.

ACLOCK.BA displays the day of the week, date, time and five alarm settings on the Tandy 200 screen. At any time, any of the alarms can be set or cancelled without disturbing the operation of the program.

Why have five alarms? Perhaps an individual needs to take medication several times daily. When required, the

laptop can sound an audible alarm and indicate which treatment is due.

The program begins by asking for the time for the first alarm to be set. Times should be entered in 24-hour format -- 2:15 PM is alarm time of 14:15:00. It's important to include the seconds.

If the alarm's not to be used, simply press return. Next, the times for alarms two through five are requested -- handle these as before.

After the alarms are set, the time presentation panel will appear. Whenever an alarm time occurs, the computer will play a short tune for 15 seconds, or until the space bar is pressed.

The alarms are repeating -- that is, they'll sound every 24 hours, whenever the correct time comes around. Of course, ACLOCK.BA must be kept running the whole time!

To change an alarm time, press R at the time presentation panel. Press Esc to return to the menu. Also, if function key eight is set to the default MENU, pressing F8 will exit ACLOCK.BA.

Listing of ACLOCK.BA, an Appointment Clock.

```
1000 'CLOCK PROGRAM - APPOINTMENT
      REMINDER
1010 'PROGRAMMED FOR THE TANDY 200 IN
      MAY 85
1020 'BY DR. LEO FINKELSTEIN, JR.
1030 CLS
1040 T=206
1050 PRINT "SET ALARMS--E.G. 09:34:15"
1060 FOR X=1 TO 5
1070 PRINT "Alarm";X
1080 INPUT A$(X)
1090 IF A$(X)<>" " THEN INPUT "PURPOSE";
      P$(X)
      :IF LEN(P$(X))>14 THEN INPUT P$(X)
1100 IF A$(X)=" " THEN A$(X)="DISABLED"
1110 NEXT X
1120 CLS
1130 PRINT @481,DAYS$
1140 PRINT @510,DATE$
1150 PRINT @520,STRING$(40,"=")
1160 PRINT @561,"TO CHANGE AN ALARM
      SETTING, PRESS <R>"
1170 PRINT @49,"* * * LOCAL TIME * * *"
1180 H$=LEFT$(TIME$,2)
```

```
1190 M$=MID$(TIME$,4,2)
1200 S$=RIGHT$(TIME$,2)
1210 PRINT @80,STRING$(40,231)
1220 'PRINT TIME
1230 PRINT @126,H$" hours "M$" minutes"
      S$" seconds"
1240 PRINT @160,STRING$(40,232)
1250 FOR X=1 TO 5
1260 T=T+40
1270 PRINT @T,"Alarm"X-"P$(X)"-"A$(X)
1280 IF A$(X)=TIME$ THEN GOTO 1350
1290 NEXT X
1300 T=206
1310 Y$=INKEY$
1320 IF Y$="" THEN 1180
1325 IF Y$="M" OR Y$=CHR$(27) THEN MENU
1330 IF Y$="R" OR Y$="r" THEN GOSUB
      1500 ELSE 1180
1340 GOTO 1120
1350 CLS
1360 PRINT @322,"ALARM TIME REACHED"
1370 FOR X=1000 TO 16000 STEP 1000
1380 SOUND X,10
1390 NEXT X
1400 FOR X=16000 TO 1000 STEP -1000
1410 SOUND X,10
1420 NEXT X
1430 K$=INKEY$
1440 IF K$="" THEN 1450 ELSE 1460
1450 C=C+1
1460 CLS
1470 C=0
1480 T=206
1490 GOTO 1120
1500 CLS
1510 FOR X=1 TO 5
1520 PRINT "ALARM ";X;"-";P$(X);
      "-";A$(X)
1530 NEXT X
1540 PRINT
1550 INPUT "ENTER NUMBER OF ALARM TO
      CHANGE";Q
1560 A$(Q)=" "
1570 P$(Q)=" "
1580 PRINT
1590 PRINT "NEW ENTRY (E.G., 14:20:00)
      FOR ALARM ";Q
1600 INPUT A$(Q)
1610 IF A$(Q)<>" " THEN INPUT "PURPOSE";
      P$(Q)
      :IF LEN(P$(Q))>14 THEN INPUT P$(Q)
1620 IF A$(Q)=" " THEN A$(Q)="DISABLED"
1630 CLS
1640 GOTO 1120
```

BEHIND THE STAGE

A few readers have asked how we manage to put slashes through the zeros in PPR program listings. It's pretty easy with many dot-matrix printers, but trickier with daisy-wheel printers.

We use a Radio Shack DWP-410 for our program listings, at it so happens that the printer uses the code Ctrl-H n for a backspace. Ctrl-H is ASCII code 8; the n refers to the number of 120ths of an inch the printer is to backspace.

Since we print listings at 12 characters per inch (cpi), each character is 10 120ths wide. That's ASCII code 10, or Ctrl-J. (Were we printing at 10 cpi, it would be Ctrl-L.) So, to print a zero, back up the correct distance, and print a shash, we use the codes

0 Ctrl-H Ctrl-J /

Voila! 0 and Ø. Of course, with different printers the codes will vary, but the principle's the same.

Alan L. Zeichick
Editor

PORTABLE PROGRAM REVIEW
Highland Mill
Camden, ME 04843

FIRST CLASS U.S. POSTAGE PAID CAMDEN, ME. PERMIT NO. 5
--

SUBSCRIBER: PLEASE ADVISE IF ABOVE ADDRESS IS INCORRECT

PORTABLE PROGRAM REVIEW is published monthly and copyright (c) 1985, all rights reserved, by Camden Communications Inc., a Maine corporation with offices at Highland Mill (P.O. Box 250), Camden, Maine 04843, James S. Povec, President. Contents of this publication may not be reproduced in whole or in part unless expressly authorized in writing by the publisher. Tandy and Radio Shack are registered trademarks of Tandy Corporation. U.S. Subscription \$29.97. Please address all subscriptions and inquiries to Nancy Wight, Circulation Director, (207) 236-4365, CompuServe ID 76703,372. ISSN: 0883-7295.